
ΑΣΚΗΣΗ 2: ΠΡΟΤΥΠΑ ΚΑΙ ΕΝΕΡΓΕΙΕΣ
ΔΟΜΙΚΑ ΣΤΟΙΧΕΙΑ ΤΟΥ CLIPS
Σύμβολα (Symbols): οποιαδήποτε ακολουθία χαρακτήρων που δεν περιλαμβάνει τους χαρακτήρες < | \ & ( ) ;   και επιπλέον δεν ξεκινά με τους χαρακτήρες  $ ? + -

Π.χ.: is-duck, door_32 

Αλφαριθμητικά (Strings): μηνύματα μέσα σε διπλά εισαγωγικά.
Π.χ.: “This is a program”  , “The phone number is 123456”.
Αριθμοί: Το CLIPS υποστηρίζει ακέραιους (μεγάλους) και αριθμούς κινητής υποδιαστολής (διπλής ακρίβειας).  

Π.χ.: 23, -45, 3.34, 4e10, 5E8
Μεταβλητές: 
Οι μονότιμες μεταβλητές στο CLIPS παίρνουν σαν ονόματα σύμβολα που ξεκινούν με τον χαρακτήρα ? . Σαν τιμή παίρνουν ένα αλφαριθμητικό, ένα σύμβολο ή έναν αριθμό. Π.χ. η ?var1 μπορεί να πάρει τιμή ένα από τα symbol1, 26, -88 .
Οι μεταβλητές πολλαπλών τιμών παίρνουν σαν ονόματα σύμβολα που ξεκινούν με τον χαρακτήρα $? . Σαν τιμή παίρνουν ένα ή περισσότερα αλφαριθμητικά, σύμβολα ή αριθμούς. Π.χ. η $?var1 μπορεί να πάρει τιμή τα (mon tue wed thu) ή (name Nikos Nikou). 
ΓΕΓΟΝΟΤΑ ΜΑΚΡΑΣ ΔΙΑΡΚΕΙΑΣ ΖΩΗΣ (Persistent facts)
Πολλές φορές ένα σύνολο γεγονότων το χρησιμοποιώ πολλές φορές. Αντί να τα γράψω πολλές φορές, χρησιμοποιώ την εντολή (deffacts). Με αυτόν τον τρόπο κατασκευάζω γεγονότα τα οποία δημιουργούνται ξανά κάθε φορά που εκτελώ την εντολή (reset). Για παράδειγμα η εντολή 
(deffacts startup (animal dog) (animal duck) (animal haddock)) 

θα προσθέτει τα τρία γεγονότα στην λίστα γεγονότων κάθε φορά που θα γίνεται reset στο σύστημα. 
Προσθέστε με εντολή (deffacts) τα παρακάτω γεγονότα στην λίστα γεγονότων. 

(animal dog) (animal cat) (animal duck) (animal turtle) (warm-blooded dog) (warm-blooded cat) (warm-blooded duck) (lays-eggs duck) (lays-eggs turtle) (child-of dog puppy) (child-of cat kitten) (child-of turtle hatchling)
ΤΑΥΤΟΠΟΙΗΣΗ (ΜATCHING)
Με την χρήση του χαρακτήρα ? μπορώ να ταιριάσω την συνθήκη ενός κανόνα με περισσότερα από ένα γεγονότα.  Για παράδειγμα ο κανόνας 
(defrule animal

  (animal ?)

  =>

  (printout t "animal found" crlf))
θα έχει τα εξής αποτελέσματα όταν τρέξει:
CLIPS>(run)
Animal found
Animal found
Animal found
Animal found
CLIPS>
Αυτό δείχνει ότι ο κανόνας έχει πυροδοτηθεί 4 φορές, μια για κάθε γεγονός που ταυτοποιείται με το πρότυπο (animal ?). Το σύμβολο ? αντικαθιστά οποιοδήποτε σύμβολο. Μπορώ να έχω περισσότερα ? σε ένα πρότυπο, αλλά το πρώτο σύμβολο του προτύπου δεν μπορεί να είναι ?. Έτσι, το(child-of ? ?) είναι αποδεκτό, ενώ το (? ? hatchling) δεν είναι.
ΜΕΤΑΒΛΗΤΕΣ ΣΤΑ ΠΡΟΤΥΠΑ
Χρησιμοποιώντας μεταβλητές στους κανόνες μπορούμε να χρησιμοποιούμε τις τιμές των μεταβλητών κάθε φορά που αυτοί εκτελούνται. Για παράδειγμα:
(defrule list-animals
(animal ?name)
=>
(printout t ?name " found" crlf))
Αυτό θα έχει το εξής αποτέλεσμα:

CLIPS>(run)
turtle found
duck found
cat found
dog found
CLIPS>
Ο κανόνας πυροδοτήθηκε από 4 γεγονότα και κάθε φορά η μεταβλητή ?name παίρνει την τιμή του συμβόλου που αναπαριστά στο πρότυπο. Έτσι στο τελευταίο μέρος του κανόνα μπορεί να εκτυπωθεί. Μπορώ σε έναν κανόνα να έχω περισσότερα από ένα πρότυπα:
(defrule mammal

  (animal ?name)

  (warm-blooded ?name)

  (not (lays-eggs ?name))

  =>

  (assert (mammal ?name))

  (printout t ?name " is a mammal" crlf))
Η συνάρτηση (not) είναι η λογική άρνηση. Αυτός ο κανόνας θα έχει σαν αποτέλεσμα: 

CLIPS>(run)
cat is a mammal
dog is a mammal
CLIPS>
Στην συνέχεια πληκτρολογήστε τον κανόνα: 
(defrule mammal2

  (mammal ?name)

  (child-of ?name ?young)

  =>

  (assert (mammal ?young))

  (printout t ?young " is a mammal" crlf))
Αφού τον τρέξετε, ελέγξτε την λίστα γεγονότων:
CLIPS>(run)
kitten is a mammal
puppy is a mammal
CLIPS>
ΔΙΑΓΡΑΦΗ  ΓΕΓΟΝΟΤΩΝ ΜΕΣΑ ΑΠΟ ΚΑΝΟΝΕΣ
Μπορείτε να διαγράψετε γεγονότα μέσα από κανόνες, συνδέοντάς τα με μεταβλητές: 
(defrule remove-mammals

  ?fact <- (mammal ?)

  =>

  (printout t "retracting " ?fact crlf)

  (retract ?fact))
Στην συνθήκη αυτού του κανόνα η μεταβλητή ?fact παίρνει σαν τιμή διαδοχικά τον αριθμό κάθε γεγονότος που ταιριάζει με το πρότυπο (mammal ?). Αυτό συμβολίζεται με το τόξο (<-). Αν τρέξετε τον κανόνα, το αποτέλεσμα θα είναι:
CLIPS>(run)
retracting <Fact-13>
retracting <Fact-14>
retracting <Fact-15>
retracting <Fact-16>
CLIPS>
Όλα τα γεγονότα με τα θηλαστικά έχουν αφαιρεθεί. 
ΑΡΙΘΜΗΤΙΚΟΙ ΚΑΙ ΛΟΓΙΚΟΙ ΤΕΛΕΣΤΕΣ
Περισσότερες από μία συνθήκες σε έναν κανόνα συνδέονται με τον τελεστή AND (πρέπει όλοι να ικανοποιούνται προκειμένου να πυροδοτηθεί ο κανόνας). Εάν θέλω να συνδέσω τις συνθήκες με τον τελεστή OR, συντάσσω τον κανόνα ως εξής:
(defrule take-umbrella

  (or (weather raining)

      (weather snowing))

  =>

  (assert (umbrella required)))
Αυτό σημαίνει ότι «εάν ή βρέχει ή χιονίζει, τότε πάρε ομπρέλα». 

Προσοχή: Στην CLIPS οι λογικοί και οι αριθμητικοί τελεστές μπαίνουν πριν από τους τελεστέους. 

Άν θέλω να υπολογίσω το άθροισμα 5 + 7, θα γράψω (+ 5 7). 
Εξετάστε τα παρακάτω παραδείγματα που δείχνουν την πρόσθεση, την αφαίρεση, τον πολλαπλασιασμό και την διαίρεση.
CLIPS>(+ 5 7)

12

CLIPS>(- 5 7)

-2

CLIPS>(* 5 7)

35

CLIPS>(/ 5 7)

0.7142857142857143

CLIPS>
Γράψτε την παράσταση 10+4*19-35/12 σε CLIPS, και ελέγξτε ότι παίρνετε το αποτέλεσμα 83.0833. 
ΕΙΣΟΔΟΣ ΔΕΔΟΜΕΝΩΝ
Για είσοδο δεδομένων από τον χρήστη, το CLIPS χρησιμοποιεί την συνάρτηση (read). 
Πληκτρολογήστε την εντολή (assert (user-input (read))). 
Μόλις πατήσετε το ENTER θα πρέπει να εισάγετε κάτι (οτιδήποτε) για να ολοκληρωθεί η εντολή. Εάν ελέγξετε τα γεγονότα θα δείτε ένα καινούριο γεγονός με την είσοδό σας στο δεύτερο μέρος του.

Πληκτρολογήστε τον κανόνα
(defrule what-is-child

        (animal ?name)

        (not (child-of ?name ?))

=>

        (printout t "What do you call the child of a " ?name "?")

        (assert (child-of ?name (read))))
Όταν τον τρέξετε, το σύστημα θα ζητήσει ένα όνομα για ένα μικρό ζώο (για όποιο μικρό ζώο δεν έχει το όνομα ήδη στην λίστα γεγονότων). Με το όνομα που θα δώσετε θα προσθέσει ένα ακόμα γεγονός στην λίστα γεγονότων.

Στο παράδειγμα προβλημάτων εκκίνησης του αυτοκινήτου στο προηγούμενο εργαστήριο, θα μπορούσατε να χρησιμοποιήσετε κανόνες της μορφής: 
(defrule are-lights-working

        (not (lights-working ?))

=>

        (printout t "Are the car's lights working (yes or no)?")

        (assert (lights-working (read))))

